

NAG Library Routine Document

F11DDF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

F11DDF solves a system of linear equations involving the preconditioning matrix corresponding to SSOR applied to a real sparse nonsymmetric matrix, represented in co-ordinate storage format.

2 Specification

```

SUBROUTINE F11DDF(TRANS, N, NNZ, A, IROW, ICOL, RDIAG, OMEGA, CHECK, Y,
1                X, IWORK, IFAIL)
    INTEGER          N, NNZ, IROW(NNZ), ICOL(NNZ), IWORK(2*N+1), IFAIL
    double precision A(NNZ), RDIAG(N), OMEGA, Y(N), X(N)
    CHARACTER*1      TRANS, CHECK

```

3 Description

F11DDF solves a system of linear equations

$$Mx = y, \quad \text{or} \quad M^T x = y,$$

according to the value of the parameter TRANS, where the matrix

$$M = \frac{1}{\omega(2 - \omega)}(D + \omega L)D^{-1}(D + \omega U)$$

corresponds to symmetric successive-over-relaxation (SSOR) (see Young (1971)) applied to a linear system $Ax = b$, where A is a real sparse nonsymmetric matrix stored in co-ordinate storage (CS) format (see Section 2.1.1 in the F11 Chapter Introduction).

In the definition of M given above D is the diagonal part of A , L is the strictly lower triangular part of A , U is the strictly upper triangular part of A , and ω is a user-defined relaxation parameter.

It is envisaged that a common use of F11DDF will be to carry out the preconditioning step required in the application of F11BEF to sparse linear systems. For an illustration of this use of F11DDF see the example program given in Section 9. F11DDF is also used for this purpose by the Black Box routine F11DEF.

4 References

Young D (1971) *Iterative Solution of Large Linear Systems* Academic Press, New York

5 Parameters

1: TRANS – CHARACTER*1 *Input*

On entry: specifies whether or not the matrix M is transposed.

TRANS = 'N'
 $Mx = y$ is solved.

TRANS = 'T'
 $M^T x = y$ is solved.

Constraint: TRANS = 'N' or 'T'.

- 2: N – INTEGER *Input*
On entry: n , the order of the matrix A .
Constraint: $N \geq 1$.
- 3: NNZ – INTEGER *Input*
On entry: the number of nonzero elements in the matrix A .
Constraint: $1 \leq \text{NNZ} \leq N^2$.
- 4: A(NNZ) – **double precision** array *Input*
On entry: the nonzero elements in the matrix A , ordered by increasing row index, and by increasing column index within each row. Multiple entries for the same row and column indices are not permitted. The routine F11ZAF may be used to order the elements in this way.
- 5: IROW(NNZ) – INTEGER array *Input*
6: ICOL(NNZ) – INTEGER array *Input*
On entry: the row and column indices of the nonzero elements supplied in A .
Constraints:
IROW and ICOL must satisfy the following constraints (which may be imposed by a call to F11ZAF):
 $1 \leq \text{IROW}(i) \leq N$ and $1 \leq \text{ICOL}(i) \leq N$, for $i = 1, 2, \dots, \text{NNZ}$;
 $\text{IROW}(i-1) < \text{IROW}(i)$ or $\text{IROW}(i-1) = \text{IROW}(i)$ and $\text{ICOL}(i-1) < \text{ICOL}(i)$, for $i = 2, 3, \dots, \text{NNZ}$.
- 7: RDIAG(N) – **double precision** array *Input*
On entry: the elements of the diagonal matrix D^{-1} , where D is the diagonal part of A .
- 8: OMEGA – **double precision** *Input*
On entry: the relaxation parameter ω .
Constraint: $0.0 < \text{OMEGA} < 2.0$.
- 9: CHECK – CHARACTER*1 *Input*
On entry: specifies whether or not the CS representation of the matrix M should be checked.
CHECK = 'C'
Checks are carried on the values of N , NNZ , IROW , ICOL and OMEGA .
CHECK = 'N'
None of these checks are carried out.
See also Section 8.2.
Constraint: CHECK = 'C' or 'N'.
- 10: Y(N) – **double precision** array *Input*
On entry: the right-hand side vector y .
- 11: X(N) – **double precision** array *Output*
On exit: the solution vector x .
- 12: IWORK($2 \times N + 1$) – INTEGER array *Workspace*

13: IFAIL – INTEGER

Input/Output

On entry: IFAIL must be set to 0, -1 or 1. If you are unfamiliar with this parameter you should refer to Section 3.3 in the Essential Introduction for details.

On exit: IFAIL = 0 unless the routine detects an error (see Section 6).

For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, if you are not familiar with this parameter, the recommended value is 0. **When the value -1 or 1 is used it is essential to test the value of IFAIL on exit.**

6 Error Indicators and Warnings

If on entry IFAIL = 0 or -1, explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

IFAIL = 1

On entry, TRANS $\neq$ 'N' or 'T',
or CHECK $\neq$ 'C' or 'N'.

IFAIL = 2

On entry, $N < 1$,
or NNZ < 1 ,
or NNZ $> N^2$,
or OMEGA lies outside the interval (0.0, 2.0),

IFAIL = 3

On entry, the arrays IROW and ICOL fail to satisfy the following constraints:

$1 \leq \text{IROW}(i) \leq N$ and $1 \leq \text{ICOL}(i) \leq N$, for $i = 1, 2, \dots, \text{NNZ}$;

$\text{IROW}(i-1) < \text{IROW}(i)$ or $\text{IROW}(i-1) = \text{IROW}(i)$ and $\text{ICOL}(i-1) < \text{ICOL}(i)$, for $i = 2, 3, \dots, \text{NNZ}$.

Therefore a nonzero element has been supplied which does not lie in the matrix A , is out of order, or has duplicate row and column indices. Call F11ZAF to reorder and sum or remove duplicates.

IFAIL = 4

On entry, the matrix A has a zero diagonal element. The SSOR preconditioner is not appropriate for this problem.

7 Accuracy

If TRANS = 'N' the computed solution x is the exact solution of a perturbed system of equations $(M + \delta M)x = y$, where

$$|\delta M| \leq c(n)\epsilon|D + \omega L||D^{-1}||D + \omega U|,$$

$c(n)$ is a modest linear function of n , and ϵ is the *machine precision*. An equivalent result holds when TRANS = 'T'.

8 Further Comments

8.1 Timing

The time taken for a call to F11DDF is proportional to NNZ.

8.2 Use of CHECK

It is expected that a common use of F11DDF will be to carry out the preconditioning step required in the application of F11BEF to sparse linear systems. In this situation F11DDF is likely to be called many times with the same matrix M . In the interests of both reliability and efficiency, you are recommended to set CHECK to 'C' for the first of such calls, and to 'N' for all subsequent calls.

9 Example

This example solves a sparse linear system of equations:

$$Ax = b,$$

using RGMRES with SSOR preconditioning.

The RGMRES algorithm itself is implemented by the reverse communication routine F11BEF, which returns repeatedly to the calling program with various values of the parameter IREVCM. This parameter indicates the action to be taken by the calling program.

If IREVCM = 1, a matrix-vector product $v = Au$ is required. This is implemented by a call to F11XAF.

If IREVCM = -1, a transposed matrix-vector product $v = A^T u$ is required in the estimation of the norm of A . This is implemented by a call to F11XAF.

If IREVCM = 2, a solution of the preconditioning equation $Mv = u$ is required. This is achieved by a call to F11DDF.

If IREVCM = 4, F11BEF has completed its tasks. Either the iteration has terminated, or an error condition has arisen.

For further details see the routine document for F11BEF.

9.1 Program Text

```
*      F11DDF Example Program Text
*      Mark 20 Revised. NAG Copyright 2001.
*      .. Parameters ..
      INTEGER          NIN, NOUT
      PARAMETER        (NIN=5,NOUT=6)
      INTEGER          NMAX, LA, LIWORK, LWORK
      PARAMETER        (NMAX=1000,LA=10000,LIWORK=2*NMAX+1,LWORK=10000)
*      .. Local Scalars ..
      DOUBLE PRECISION ANORM, OMEGA, SIGMAX, STPLHS, STPRHS, TOL
      INTEGER          I, IFAIL, IREVCM, ITERM, ITN, LWNEED, LWREQ, M,
+      MAXITN, MONIT, N, NNZ
      CHARACTER        CKDDF, CKXAF, NORM, PRECON, TRANS, WEIGHT
      CHARACTER*8       METHOD
*      .. Local Arrays ..
      DOUBLE PRECISION A(LA), B(NMAX), RDIAG(NMAX), WGT(NMAX),
+      WORK(LWORK), X(NMAX)
      INTEGER          ICOL(LA), IROW(LA), IWORK(LIWORK)
*      .. External Subroutines ..
      EXTERNAL         F11BDF, F11BEF, F11BFF, F11DDF, F11XAF
*      .. Intrinsic Functions ..
      INTRINSIC        MAX
*      .. Executable Statements ..
      WRITE (NOUT,*) 'F11DDF Example Program Results'
      WRITE (NOUT,*)
*      Skip heading in data file
      READ (NIN,*)
*
*      Read algorithmic parameters
*
      READ (NIN,*) N
      IF (N.LE.NMAX) THEN
        READ (NIN,*) NNZ
        READ (NIN,*) METHOD
```

```

      READ (NIN,*) PRECON, NORM, ITERM
      READ (NIN,*) M, TOL, MAXITN
      READ (NIN,*) ANORM, SIGMAX
      READ (NIN,*) OMEGA
*
*      Check size of workspace
*
      LWREQ = MAX(N*(M+3)+M*(M+5)+101,7*N+100,(2*N+M)*(M+2)+N+100,
+          10*N+100)
      IF (LWORK.LT.LWREQ) THEN
          WRITE (NOUT,*) 'LWORK must be at least', LWREQ
          GO TO 160
      END IF
*
*      Read the matrix A
*
      DO 20 I = 1, NNZ
          READ (NIN,*) A(I), IROW(I), ICOL(I)
20      CONTINUE
*
*      Read right-hand side vector b and initial approximate solution x
*
      READ (NIN,*) (B(I),I=1,N)
      READ (NIN,*) (X(I),I=1,N)
*
*      Call F11BDF to initialize solver
*
      WEIGHT = 'N'
      MONIT = 0
      IFAIL = 1
      CALL F11BDF(METHOD,PRECON,NORM,WEIGHT,ITERM,N,M,TOL,MAXITN,
+          ANORM,SIGMAX,MONIT,LWNEED,WORK,LWORK,IFAIL)
      IF (IFAIL.NE.0) THEN
          WRITE (NOUT,99999) IFAIL
          GO TO 160
      END IF
*
*      Calculate reciprocal diagonal matrix elements if necessary
*
      IF (PRECON.EQ.'P' .OR. PRECON.EQ.'p') THEN
*
          DO 40 I = 1, N
              IWORK(I) = 0
40          CONTINUE
*
          DO 60 I = 1, NNZ
              IF (IROW(I).EQ.ICOL(I)) THEN
                  IWORK(IROW(I)) = IWORK(IROW(I)) + 1
                  IF (A(I).NE.0.DO) THEN
                      RDIAG(IROW(I)) = 1.DO/A(I)
                  ELSE
                      WRITE (NOUT,*) 'Matrix has a zero diagonal element'
                      GO TO 140
                  END IF
              END IF
60          CONTINUE
*
          DO 80 I = 1, N
              IF (IWORK(I).EQ.0) THEN
                  WRITE (NOUT,*) 'Matrix has a missing diagonal element'
                  GO TO 140
              END IF
              IF (IWORK(I).GE.2) THEN
                  WRITE (NOUT,*)
+                      'Matrix has a multiple diagonal element'
                  GO TO 140
              END IF
80          CONTINUE
*
          END IF
*

```

```

*      Call F11BEF to solve the linear system
*
      IREVCM = 0
      CKXAF = 'C'
      CKDDF = 'C'
*
100    CONTINUE
*
      IFAIL = 0
      CALL F11BEF(IREVCM,X,B,WGT,WORK,LWORK,IFAIL)
*
      IF (IREVCM.EQ.1) THEN
*
*         Compute matrix-vector product
*
          TRANS = 'N'
          CALL F11XAF(TRANS,N,NNZ,A,IROW,ICOL,CKXAF,X,B,IFAIL)
          CKXAF = 'N'
          GO TO 100
*
      ELSE IF (IREVCM.EQ.-1) THEN
*
*         Compute transposed matrix-vector product
*
          TRANS = 'T'
          CALL F11XAF(TRANS,N,NNZ,A,IROW,ICOL,CKXAF,X,B,IFAIL)
          CKXAF = 'N'
          GO TO 100
*
      ELSE IF (IREVCM.EQ.2) THEN
*
*         SSOR preconditioning
*
          TRANS = 'N'
          CALL F11DDF(TRANS,N,NNZ,A,IROW,ICOL,RDIAG,OMEGA,CKDDF,X,B,
+              IWORK,IFAIL)
          CKDDF = 'N'
          GO TO 100
*
      ELSE IF (IREVCM.EQ.4) THEN
*
*         Termination
*
          IFAIL = 0
          CALL F11BFF(ITN,STPLHS,STPRHS,ANORM,SIGMAX,WORK,LWORK,IFAIL)
*
          WRITE (NOUT,'(A,I10,A)') ' Converged in', ITN, ' iterations'
          WRITE (NOUT,'(A,1P,E16.3)') ' Matrix norm =', ANORM
          WRITE (NOUT,'(A,1P,E16.3)') ' Final residual norm =', STPLHS
          WRITE (NOUT,*)
*
*         Output x
*
          WRITE (NOUT,*) '          X'
          DO 120 I = 1, N
              WRITE (NOUT,'(1X,1P,E16.4)') X(I)
120      CONTINUE
*
      END IF
*
140    CONTINUE
*
      END IF
160    CONTINUE
*
99999 FORMAT (1X,/1X,' ** F11BDF returned with IFAIL = ',I5)
      END

```

9.2 Program Data

F11DDF Example Program Data

```

5          N
16         NNZ
'RGMRES'   METHOD
'P' 'I' 1   PRECON, NORM, ITERM
2 1.D-10 1000 M, TOL, MAXITN
0.DO 0.DO   ANORM, SIGMAX
1.1DO      OMEGA

2. 1 1
1. 1 2
-1. 1 4
-3. 2 2
-2. 2 3
1. 2 5
1. 3 1
5. 3 3
3. 3 4
1. 3 5
-2. 4 1
-3. 4 4
-1. 4 5
4. 5 2
-2. 5 3
-6. 5 5      A(I), IROW(I), ICOL(I), I=1,...,NNZ
0. -7. 33.
-19. -28.     B(I), I=1,...,N
0. 0. 0.
0. 0.         X(I), I=1,...,N

```

9.3 Program Results

F11DDF Example Program Results

```

Converged in      12 iterations
Matrix norm =     1.200E+01
Final residual norm = 3.841E-09

```

```

      X
1.0000E+00
2.0000E+00
3.0000E+00
4.0000E+00
5.0000E+00

```
